



HØGSKOLEN
I SØR-TRØNDELAG

HØGSKOLEN I SØR-TRØNDELAG


Avdeling for informatikk og e-læring - AITeL

Kandidatnr:	
Eksamensdato:	20.mai 2005
Varighet:	3 timer
Fagnummer:	LO116D
Fagnavn:	Programmering i Visual Basic
Klasse(r):	FU
Studiepoeng:	6
Faglærer:	Svend Andreas Horgen tlf: 73 55 92 69
Kursassistent:	Siri Wæhre Lien tlf: 73 55 91 54
Hjelpemidler:	Alle skriftlige hjelpemidler
Oppgavesettet består av:	3 oppgaver og totalt 7 sider (inkludert forsiden). Siste side har to vedlegg, merket Vedlegg A og Vedlegg B. Det står i oppgaveteksten hvor disse kan brukes.
Viktig: Planlegg tiden godt. Det er ikke gitt at deloppgavene vektet likt. Dersom noe virker uklart i oppgavene, må du selv gjøre de nødvendige antakelser og få med disse i besvarelsen.	
Lykke til!	

Oppgave 1 – Variabler og kontrollstrukturer (30%)

Du har sikkert sett annonser for bruktbiler. Du får gjerne opplyst årsmodell, det vil si når bilen var ny, og kjørelengde, som er hvor langt bilen har gått totalt. Noen kjører 14.000 km, andre 25.000 km per år. Denne informasjonen står ikke i bilannonser.

Du skal nå lage et program som kan regne ut hvor langt en bil har kjørt i gjennomsnitt per år. For enkelhets skyld skal du ikke regne i måneder, men kun i hele år. Altså er en bil som ble kjøpt i 2004, 1 år gammel siden vi nå er i 2005. Programmet bør være brukbart i mange år framover. Anta at bruktbiler er minst 1 år gamle.



Merk: Nyttige kommandoer som kan brukes i denne oppgaven er å finne i **Vedlegg A**

- Forklar hvordan du vil gå fram for å fylle komboboksen (årsmodell) med innhold. Det skal være mulig å velge modeller 20 år bakover i tid fra dagens dato.

Svar: Bruk en for-løkke i `form_Load`. Step -1 fører til negativt inkrement.

```
Private Sub Form1_Load(...) Handles MyBase.Load
    Dim i As Integer
    Dim naa As Integer = Date.Now.Year
    For i = naa - 1 To naa - 1 - 20 Step -1
        cboModell.Items.Add(i)
    Next
End Sub
```

- Lag koden som finner gjennomsnittlig årlig kjørelengde når knappen "Beregn årlig kjørelengde" trykkes. Svaret skal komme i det første tekstfeltet rett ved siden av "Årlig kjørelengde". Sjekk også at den kjørelengden som skrives inn faktisk er et tall. Hvis ikke, så opplys om dette i en meldingsboks.


Svar:

```
Private Sub btnBeregn_Click(ByVal ... ) Handles btnBeregn.Click
    Dim lengde As String = txtLengde.Text
    Dim naa As Integer = Date.Now.Year
    Dim alder As Integer
    Dim snittlengde As String

    alder = naa - cboModell.SelectedItem

    If IsNumeric(lengde) Then
        snittlengde = lengde / alder
    Else
        MsgBox("kjørelengden må være et tall")
    End If
    txtSnitt.Text = snittlengde
End Sub
```

- c. Du ser et ekstra tekstfelt til høyre for feltet med gjennomsnittsverdien du fant i oppgave b. Dette feltet skal fargelegges avhengig av kjørelengdens verdi. For kjørelengder under 10.000 km skal feltet bli blått, for lengder mellom 10.000 og 20.000 km skal feltet bli gult, og for lengder over 20.000 km skal feltet bli rødt. Neste bilde viser et eksempel hvor feltet blir blått.



Svar: Legg til følgende linjer

```
If snittlengde < 10000 Then
    txtStatus.BackColor = System.Drawing.Color.Blue
ElseIf snittlengde < 20000 Then
    txtStatus.BackColor = System.Drawing.Color.Yellow
Else
    txtStatus.BackColor = System.Drawing.Color.Red
End If
```

Oppgave 2 – Matriser og funksjoner (40%)

Du vil lage ditt eget bibliotek over filmer, og starter derfor med å lagre navnet på en rekke DVD-filmer i en matrise. Du antar at du vil få bruk for funksjonalitet spredd over flere knapper, og derfor lager du en global matrise og variabelen `nesteLedigePlass` som holder orden på hvilket element neste film kan lagres i.

```
Private FilmMatrise(999) As String  
Private nesteLedigePlass As Integer 'har 0 i utgangspunktet
```


- Du skal nå lage koden som ligger bak knappen ”Registrer film” som vist i figuren under. Den skal legge navnet på den filmen som skrives inn tekstfeltet, i neste ledige matriseelement. Vær obs på å ikke overskrive andre filmer som ligger i matrisen fra før.



Svar:

```
Private Sub btnRegistrerFilm_Click(...) Handles ...  
    If nesteLedigePlass <= 999 Then  
        FilmMatrise(nesteLedigePlass) = txtFilmNavn.Text  
        nesteLedigePlass = nesteLedigePlass + 1  
    End If  
    txtFilmNavn.Clear() 'ikke nødvendig  
    txtFilmNavn.Focus() 'ikke nødvendig  
End Sub
```

- b. Lag kode som skriver ut de registrerte filmene i en listeboks, som vist i neste figur.



Svar: Bruker With for å knytte referanse til objektet bare en gang.

```
Private Sub btnSkrivUt_Click(ByVal ... ) Handles btnSkrivUt.Click
    Dim i As Integer
    With lstUtskrift.Items
        For i = 0 To nesteLedigePlass - 1
            .Add(FilmMatrise(i))
        Next i
    End With
End Sub
```

- c. Lag en **funksjon** som mottar navnet på en film som argument, og søker gjennom matrisen `FilmMatrise`. Dersom filmen blir funnet, skal nøkkelverdien til matriseelementet navnet ble funnet i, returneres. Hvis ikke filmen blir funnet, skal -1 returneres.

Svar:

```
Private Function finnFilm(ByVal navn As String) As Integer
    Dim nr, teller As Integer
    Dim funnet As Boolean = False
    nr = -1

    'Trenger ikke å gjennomgå til mer enn nesteLedigePlass
    Do While teller < nesteLedigePlass And Not funnet
        If navn = FilmMatrise(teller) Then
            nr = teller
            funnet = True
        Else
            teller = teller + 1
        End If
    Loop
    Return nr 'enten -1 eller nøkkelverdien
End Function
```


- d. Du har nå en matrise med navn på en rekke filmer. Anta at du har en parallell matrise som lagrer terningkastet du har gitt til hver film, og at den har innhold før du tar fatt på denne oppgaven. Matrisen er deklarerert slik:

```
Private TerningMatrise() As Integer
```

Til venstre i figuren under er de første elementene av matrisen `FilmMatrise` vist, og til høyre ser du de første elementene i matrisen `TerningMatrise`. Du ser for eksempel at filmene "Elling" og "The Rock" har fått terningkast 4.

The Rock	4
Butch Cassidy and the Sundance Kid	5
Fabelaktige Amelie	6
Les Miserables	5
Elling	4
Foyle's War	6

Neste figur viser meldingsboksen som kommer fram når knappen "Finn terning på film" trykkes. Du skal nå lage kode til denne knappen. Basert på det navnet som skrives inn i tekstfeltet, skal en meldingsboks vise enten at filmen ikke ble funnet, eller opplyse om hvilket terningkast filmen har fått. **Tips:** Bruk funksjonen du lagde i oppgave c.




Svar:

```
Private Sub btnFinnFilm_Click(...) Handles btnFinnFilm.Click
    Dim filmID As Integer
    filmID = finnFilm(txtFilmNavn.Text)
    If filmID >= 0 Then
        Dim terning As Integer = TerningMatrise(filmID)
        MsgBox(txtFilmNavn.Text & " fikk terningkast " & terning)
    Else
        MsgBox("Filmen ble ikke funnet")
    End If
End Sub
```

Oppgave 3 – Filbehandling og grafikk (30%)

Her ser du et enkelt grensesnitt for å registrere hva lærerne i lunsjen ved AITeL helst drikker. I **Vedlegg B** ser du tre tekstfiler, en som har navn på de som drikker Cola, en for Cola Light og en for Melk.



- a. Du skal nå lage et program som lagrer informasjonen som skrives inn på riktig tekstfil og opplyser brukeren om at registreringen var vellykket. Lagring skal bare skje dersom en drikk er valgt og et navn skrevet inn.

Tips: Ikke overskriv det som ligger på filene fra før!

Svar:

```
Dim valg As String = cboDrikke.SelectedItem
Dim navn As String = txtNavn.Text.Trim

If valg = "Cola Light" Then
    filnavn = "colalight.txt"
ElseIf valg = "Cola" Then
    filnavn = "cola.txt"
ElseIf valg = "Melk" Then
    filnavn = "melk.txt"
End If

'Åpne eller evt. opprett ny fil for skriving
FileOpen(1, filnavn, OpenMode.Append)
PrintLine(1, navn)
FileClose(1)
MsgBox("Dine drikkevaner er registrert")
```

- b. Det er interessant å få ut en oversikt over hvem som drikker hva. Grensesnittet fra oppgave a) er nå litt utvidet. Når knappen ”Vis drikkevaner” trykkes, skal informasjonen som er lagret på tekstfilene vises til høyre i skjermbildet, i en label som heter `lblVaner`. Lag koden som gjør dette. Hint: Konstanten `vbLf` gir linjeskift.



Svar:

```
' Åpne ønsket fil
FileOpen(1, "colalight.txt", OpenMode.Input)
FileOpen(2, "cola.txt", OpenMode.Input)
FileOpen(3, "melk.txt", OpenMode.Input)

' Gå gjennom filene linje for linje
lblVaner.Text = "Liker best Cola Light:" & vbLf
Do Until EOF(1)
    lblVaner.Text &= " - " & LineInput(1) + vbLf
Loop

lblVaner.Text += vbLf & "Liker best Cola:" & vbLf
Do Until EOF(2)
    lblVaner.Text &= " - " & LineInput(2) + vbLf
Loop

lblVaner.Text += vbLf & "Kalver:" & vbLf
Do Until EOF(3)
    lblVaner.Text &= " - " & LineInput(3) + vbLf
Loop
' Lukker alle filene på engang
FileClose(1, 2, 3)
```

- c. Her ser du prosedyren `LagSoyle` som tegner et rektangel med høyde basert på argumentet `antall`. Siden en piksel er nokså liten, er høyden skalert opp med en faktor på 10. Argumentet `venstre` brukes for å plassere søylen på et visst sted på lerretet, og tekststrengen `tekst` plasseres under søylen. Koden ser slik ut:


```

Private Sub LagSoyle(ByVal tekst As String, _
                    ByVal antall As Integer, _
                    ByVal venstre As Integer)

    Dim g As Graphics
    g = PictureBox1.CreateGraphics
    Dim blyant As New Drawing.Pen(System.Drawing.Color.Black)
    Dim start_y As Integer = PictureBox1.Height - 50
    Dim hoyde As Integer = antall * 10
    g.DrawRectangle(blyant, venstre, start_y - hoyde, 50, hoyde)
    g.DrawLine(blyant, 0, start_y, PictureBox1.Width, start_y)
    Dim pensel As New SolidBrush(System.Drawing.Color.Blue)
    g.DrawString(tekst, Me.Font, pensel, venstre, start_y + 10)
End Sub

```

Hvordan kan du bruke prosedyren `LagSoyle` til å lage en statistikk som ser *nøyaktig* slik ut som vist nederst i neste figur? Utvid koden under knappen ”Vis drikkevaner” som du lagde i oppgave b, eller skriv de setningene som må føyes til og si hvor disse skal plasseres. Du får vite at det er 20 piksler mellom kanten av lerretet og første søyle, 10 piksler mellom hver søyle, og søylebredden skal være 50 piksler. I figuren under ser du piler som indikerer starten på koordinatsystemet slik VB bruker det, og forklaring til lerretets dimensjoner. Dette er kun til hjelp for deg, altså ikke en del av programmet.



Svar: For hver løkke som leser innholdet på en fil, må antall linjer summeres opp.

```
Dim ANTALL_light As Integer = 0
lblVaner.Text = "Liker best Cola Light" & vbCrLf
Do Until EOF(1)
    lblVaner.Text &= vbTab & LineInput(1) + vbCrLf
    ANTALL_light = ANTALL_light + 1
Loop

'Tilsvarende for Cola og melk
```

Når variablene ANTALL_light, ANTALL_cola, og ANTALL_melk har fått riktig innhold, må søylene lages. Legg merke til variabelen neste som dirigerer hvor plasseringen skal skje:

```
Dim neste As Integer = 20
LagStatistikk("Cola Light", ANTALL_light, neste)

neste = neste + 50 + 10

LagStatistikk("Cola", ANTALL_cola, neste)

neste = neste + 50 + 10

LagStatistikk("Melk", ANTALL_melk, neste)
```

Vedlegg A

Her er nyttige kommandoer og hint som kan hjelpe deg med å løse oppgavene.

- Finne årstall for inneværende år og skrive ut i en meldingsboks
`MsgBox Date.Now.Year`
- Sette bakgrunnsfarge i en tekstboks:
`TextBox1.BackColor = System.Drawing.Color.Blue`
'eventuelt Red, Yellow, Green, White, Black
- Legge til et element i en komboboks:
`ComboBox1.Items.Add("Det som skal legges til")`
- Skrive ut i en meldingsboks det elementet som er valgt i en komboboks:
`MsgBox ComboBox1.SelectedItem`

Vedlegg B

Filer med informasjon over folks drikkevaner

